

# Second Order Difference Equations

## Dowling18B.wmx

### TABLE OF CONTENTS

PREFACE .....	1
REFERENCES .....	2
SECOND ORDER DIFFERENCE EQUATIONS .....	2
PARTICULAR, COMPLEMENTARY, TOTAL SOLUTION .....	4
STABILITY OF SOLUTIONS .....	7
USING MAXIMA'S solve_rec FOR RECURRENCE RELATIONS .....	14
MULTIPLIER-ACCELERATOR MODEL OF NATIONAL INCOME .....	15
DISTINCT REAL ROOTS .....	24
DOUBLE REAL ROOTS .....	34
COMPLEX ROOTS .....	40
SUMMARY PLOT OF (a, b) PLANE SOLUTION REGIONS .....	51
ECONOMIC INSTABILITY EXAMPLE .....	52

```
(%i5) load(draw)$ set_draw_defaults(line_width=2, grid = [2,2], point_type = filled_circle,
    background_color = light_gray, draw_realpart=false)$
    fpprintprec:5$ ratprint:false$ kill(all)$
```

```
(%i1) load ("Econ2.mac");
```

```
(%o1) c:/work5/Econ2.mac
```

## 1 Preface

Dowling18B.wmxm is one of a number of wxMaxima files available in the section Economic Analysis with Maxima on my CSULB webpage. Some of the calculations done in this worksheet need to have the software file Econ2.mac loaded first. Use load ("Econ2.mac");

We use Maxima to solve a few of the problems (and draw some of the plots) in Sec. 18.2 [Second Order Difference Equations] and in Sec. 18.8 [Stability Conditions] of Ch. 18 of the supplemental text: Introduction to Mathematical Economics, 3rd ed, (Schaum's Outline Series), by Edward T. Dowling (1992, 2001), McGraw-Hill. This modestly priced text is a bargain with many worked out examples. You should compare the examples worked out "by hand" in this text with what we do using Maxima. Section numbers [x.y] refer to sections in Dowling's text.

We have supplemented Dowling's problem 18.37 (Samuelson's Multiplier-Accelerator Model) with the more extensive treatment of this model by Chiang and Wainwright (their Ch. 18, Sec. 2).

We have slightly changed some of the symbols used in particular problems. An approximate pdf translation (using Microsoft print to pdf) is available as Dowling18Bfit.pdf. This pdf file can be searched using Ctrl-F.

Ted Woollett  
<https://home.csulb.edu/~woollett/>  
Dec. 9, 2021

## 2 *References*

Fundamental Methods of Mathematical Economics, Alpha C. Chiang and Kevin Wainwright, 4th ed., 2005, McGraw-Hill (Ch. 18)

[https://en.wikipedia.org/wiki/Linear\\_difference\\_equation](https://en.wikipedia.org/wiki/Linear_difference_equation)

[https://en.wikipedia.org/wiki/Multiplier-accelerator\\_model](https://en.wikipedia.org/wiki/Multiplier-accelerator_model)

"How to Solve Difference Equations? - A Complete Video Tutorial", 35 min,  
Dr. Mostapha Kalami Heris,  
<https://www.youtube.com/watch?v=gI2JlWIVk8E>

Math. Methods for Economics, 22 min,  
<https://www.youtube.com/watch?v=u1ec0t1doAM>

## 3 *Concepts and Definitions*

We first review notational conventions which were presented in Dowling17.wmxm which dealt solely with first order difference equations.

A difference equation expresses a relationship between a dependent variable and a lagged independent variable (or variables) which changes at discrete intervals of time, for example,

$$I[t] = F( Y[t - 1] )$$

with  $F(x)$  being some arbitrary function, where  $I$  and  $Y$  are each measured at the end of each year. Investment during year 4 is some function of national income during year 3, for example:  $I[4] = F( Y[3] )$ , etc.

The order of a difference equation is determined by the greatest number of periods lagged. A first-order difference equation contains only a time lag of one period, as above.

A second-order difference equation involves a time lag of two periods, in addition to (perhaps) also a time lag of one period, etc.

The change in  $y$  as  $t$  changes from  $t$  to  $t + 1$  is called the "first difference of  $y$ ".

The FIRST DIFFERENCE of  $y$  is written :

$$\Delta y / \Delta t = \Delta y[t] = y[t + 1] - y[t] = \text{change in } y \text{ per period over the time } (t, t+1).$$

In Maxima,  $y[t]$  is a (hash) array and wxMaxima will display the symbol using subscript notation ( if `display2d` is set to true, which is the default).

```
(%i2) y[t];
```

```
(%o2) yt
```

```
(%i3) display2d;
```

```
(%o3) true
```

### 3.1 Examples of Difference Equations of Various Orders

Each of the following is a difference equation of the order indicated:

Order 2:

```
(%i4) I[t] = a * ( Y[t-1] - Y[t-2] );
```

```
(%o4) It = a ( Yt-1 - Yt-2 )
```

Order 1:

```
(%i5) Q[t] = a + b * P[t-1];
```

```
(%o5) Qt = b Pt-1 + a
```

Order 3:

$$(\%i6) \quad y[t+3] - 9*y[t+2] + 2*y[t+1] + 6*y[t] = 8;$$

$$(\%o6) \quad y_{t+3} - 9 y_{t+2} + 2 y_{t+1} + 6 y_t = 8$$

Order 1:

$$(\%i7) \quad \Delta y[t] = 5*y[t];$$

$$(\%o7) \quad \Delta y_t = 5 y_t$$

The last equation is the same as:

$$(\%i8) \quad y[t+1] - y[t] = 5*y[t];$$

$$(\%o8) \quad y_{t+1} - y_t = 5 y_t$$

which is the same as:

$$(\%i9) \quad y[t+1] = 6*y[t];$$

$$(\%o9) \quad y_{t+1} = 6 y_t$$

## 4 Linear Second Order Difference Equations

### 4.1 Total Solution: $y[t] + b1*y[t-1] + b2*y[t-2] = a$

The total solution  $y[t]$  of a second order linear difference equation is the sum of the particular solution  $y_p[t]$  and the complimentary solution  $y_c[t]$ :

$$y[t] = y_p[t] + y_c[t],$$

where  $y_c$  is the general solution of the homogeneous equation

#### 4.1.1 $y_{part}(b1,b2,a,t)$ , $y_{compl}(b1,b2,t)$ , $y_{soln}(b1,b2,a,t)$

Given the 2nd order linear difference equation

$$y[t] + b1*y[t-1] + b2*y[t-2] = a,$$

with constants  $b1, b2, a$ , the particular solution  $y_p[t]$  is returned by the Maxima function  $y_{part}(b1,b2,a,t)$  defined in `Econ2.mac`.

( $y_{part}$ ,  $y_{compl}$ , and  $y_{soln}$  are all defined in `Econ2.mac`.)

(%i10) `fundef (ypart);`

(%o10) `ypart(_b1,_b2,_a,_t):=if _a=0 then 0 else if _b1+_b2=-1 and  
_b1=-2 then  $\frac{-a}{2}t^2$  else if _b1+_b2=-1 then  $\frac{-a}{2+_b1}t$  else  $\frac{-a}{1+_b1+_b2}$`

(%i11) `ypart (b1,b2,a,t);`

(%o11)  $\frac{a}{b2+b1+1}$

(%i12) `ypart (-2, 1, 8, t);`

(%o12)  $4t^2$

`yc[t]` is given by the Maxima function `ycompl (b1, b2, t)`.

`ysoln (b1,b2,a,t)` returns `yp + yc` in terms of `%A1` and `%A2`, an "indefinite" solution for `y[t]`.

(%i13) `ycompl(-2,1,t);`

(%o13) `%A2 t+%A1`

(%i14) `ysoln (-2,1,8,t);`

(%o14)  $4t^2 + \%A2 t + \%A1$

## 5 Examples 5, 6 & 7

### 5.1 Eqn. 1: Real Roots, $b1^2 > 4*b2$

Find the particular solution and the complementary solution of the second order linear differential equation

$$y[t] - 10*y[t-1] + 16*y[t-2] = 14, \text{ in which } b1 = -10, b2 = 16, a = 14.$$

(%i15) `ypart (-10, 16, 14, t);`

(%o15) 2

(%i16) `ycompl (-10, 16, t);`

(%o16)  $\%A1 8^t + \%A2 2^t$

(%i17) `ysoln (-10, 16, 14, t);`

(%o17)  $\%A1 8^t + \%A2 2^t + 2$

## 5.2 Eqn. 2: Real Roots, $b_1^2 > 4*b_2$

Find the particular solution and the complementary solution of the second order linear differential equation

$$y[t] - 6*y[t-1] + 5*y[t-2] = 12, \text{ in which } b_1 = -6, b_2 = 5, a = 12.$$

```
(%i18) ypart (-6, 5, 12, t);
```

```
(%o18) -3 t
```

```
(%i19) ycompl (-6, 5, t);
```

```
(%o19) %A1 5^t+%A2
```

```
(%i20) ysoln (-6, 5, 12, t);
```

```
(%o20) %A1 5^t-3 t+%A2
```

## 5.3 Eqn. 3, Double Real Root, $b_1^2 = 4*b_2$

Find the particular solution and the complementary solution of the second order linear differential equation

$$y[t] - 2*y[t-1] + y[t-2] = 8, \text{ in which } b_1 = -2, b_2 = 1, a = 8.$$

```
(%i21) ypart (-2, 1, 8, t);
```

```
(%o21) 4 t^2
```

```
(%i22) ycompl (-2, 1, t);
```

```
(%o22) %A2 t+%A1
```

```
(%i23) ysoln (-2, 1, 8, t);
```

```
(%o23) 4 t^2+%A2 t+%A1
```

## 6 Example 8, Satisfying $y[0]$ & $y[1]$ Conditions

We return to Eqn. 1 above and require  $y[0] = 10$  and  $y[1] = 36$  and look for a definite solution respecting that data. We first define a Maxima function  $ys(t)$ . We need to use two single quotes (') in front of  $yindef$  here.

```
(%i24) yindef : ysoln (-10, 16, 14, t);
```

```
(yindef) %A1 8^t+%A2 2^t+2
```

```
(%i25) ys(t) := "yundef;
```

```
(%o25) ys(t) := %A1 8^t + %A2 2^t + 2
```

```
(%i26) solns : solve ([ ys(0) = 10, ys(1) = 36], [%A1, %A2]);
```

```
(solns) [[ %A1=3, %A2=5]]
```

```
(%i27) ydef : at (yundef, solns[1]);
```

```
(ydef) 3 8^t + 5 2^t + 2
```

We can check ydef by evaluating at t=0,1,2. It is easiest to define a Maxima function yd(t).

```
(%i28) yd(t) := "ydef;
```

```
(%o28) yd(t) := 3 8^t + 5 2^t + 2
```

```
(%i29) map ('yd, [0,1,2,3]);
```

```
(%o29) [10,36,214,1578]
```

Alternatively, you can use the Maxima at( ...) function to check the solution.

```
(%i30) at (ydef, t = 0);
```

```
(%o30) 10
```

```
(%i31) at (ydef, t = 1);
```

```
(%o31) 36
```

## 7 Prob. 18.16: Repeated Real Roots

Find the solution  $y[t]$  for the difference equation  
 $y[t] - 10 y[t-1] + 25 y[t-2] = 8$ , for which  $b_1 = -10$ ,  $b_2 = 25$ ,  $a = 8$ .

```
(%i32) ysoln (-10, 25, 8, t);
```

```
(%o32) %A2 t 5^t + %A1 5^t + 1/2
```

```
(%i33) grind(%)$
```

```
%A2*t*5^t+%A1*5^t+1/2$
```

## 8 Definite Solutions, ytlist(b1,b2,a,y0,y1,tmax)

### 8.1 Stability of Solutions

If the roots are real, the characteristic root with the largest magnitude (absolute value) is called the dominant root because it dominates the time path. For convergence, the absolute value of the dominant root must be less than 1.

The "characteristic equation" corresponding to the difference equation

$$y[t] + b_1 y[t-1] + b_2 y[t-2] = a$$

is

$$r^2 + b_1 r + b_2 = 0.$$

The solutions of this quadratic equation are the "characteristic roots", and can either be two real distinct roots  $r_1, r_2$ , if  $b_1^2 > 4b_2$ ,

a double real root  $r$ , if  $b_1^2 = 4b_2$ ,

or a pair of complex roots, if  $b_1^2 < 4b_2$ , which then also means  $b_2 > 0$ .

For the case of complex roots, stability is determined by the magnitude of  $k = \sqrt{b_2}$ .

If  $|k| < 1$  the solution converges, if  $|k| > 1$  the solution diverges.

## 8.2 Prob. 18.18: Distinct Real Roots: $b_1^2 > 4b_2$

Find the definite solution of

$$y[t] + 7y[t-1] + 6y[t-2] = 42, \text{ with } y[0] = 16, y[1] = -35.$$

Comment on stability.

For this problem,  $b_1 = 7, b_2 = 6, a = 42, y_0 = 16, y_1 = -35$ .

We first look at this problem the "long way", as a pedagogical exercise in Maxima.

We then use Maxima our do loop code ytlist, which is much faster.

Look at the characteristic roots with  $b_1 = 7$  and  $b_2 = 6$ :

```
(%i34) solve (r^2 + 7*r + 6);
```

```
(%o34) [r=-6,r=-1]
```

```
(%i35) yindef : ysoln ( 7, 6, 42, t);
```

```
(yindef) %A1 (-1)^t + %A2 (-6)^t + 3
```

Define a Maxima function ys(t) by placing two single quotes (') in front of yindef.

```
(%i36) ys(t) := 'yindef;
```

```
(%o36) ys(t) := %A1 (-1)^t + %A2 (-6)^t + 3
```



Find the definite solution by using the values of  $y[0]$  and  $y[1]$ .

```
(%i37) solns : solve ([ ys(0) = 16, ys(1) = -35], [%A1, %A2]);
(solns) [[ %A1=8, %A2=5 ]]
```

```
(%i38) ydef : at (yindex, solns[1]);
(ydef) 8 (-1)^t + 5 (-6)^t + 3
```

The characteristic roots are -1 and -6. The dominant root is -6. Since  $|-6| = 6 > 1$ , the time path is divergent.

Again, use two single quote marks in front of ydef to define the Maxima function yd(t).

```
(%i39) yd(t) := "ydef;
(%o39) yd(t) := 8 (-1)^t + 5 (-6)^t + 3
```

Make a list of times and a list of values of  $y[t]$ .


```
(%i41) tmax : 5$
      tL : makelist(t, t, 0.0, tmax);
(tL) [0.0, 1.0, 2.0, 3.0, 4.0, 5.0]

(%i42) ytL : map ('yd, tL), numer;
(ytL) [16.0, -35.0, 191.0, -1085.0, 6491.0, -3.8885 10^4]
```

We see oscillation between positive and negative values, each point further from the intertemporal equilibrium value  $y_e = 3$ ; explosive instability.

```
(%i43) wxdraw2d (xlabel = "t", ylabel = "y[t]", xrange = [0,tmax],yrange = [-1.1e3,1.1e3],
title = "b1 = 7, b2 = 6, a = 42, y[0] = 16, y[1] = -35",
points (tL, ytL), line_width = 1, color = red, key = "ye = 3", explicit(3, t, 0, tmax))$
```

(%t43)



Now for the faster path. A Maxima function


```
ytlist (b1, b2, a, y0, y1, tmax)
```

is defined in Econ2.mac. ytlist produces a list of values of  $y[t]$  from the given second order difference equation.

```
(%i46) tmax : 5$
tL : makelist (t, t, 0, tmax);
ytL : ytlist (7, 6, 42, 16, -35, tmax);
(tL) [0, 1, 2, 3, 4, 5]
ye = 3
(ytL) [16, -35, 191, -1085, 6491, -38885]
```

```
(%i47) wxdraw2d (xlabel = "t", ylabel = "y[t]", xrange = [0,tmax],yrange = [-1.1e3,1.1e3],
title = "b1 = 7, b2 = 6, a = 42, y[0] = 16, y[1] = -35",
points (tL, ytL), line_width = 1, color = red, key = "ye = 3", explicit(ye, t, 0, tmax))$
```

```
(%t47)
```



A look at longer times (without a plot):

```
(%i48) float ( ytolist (7, 6, 42, 16, -35, 30) );
ye=3
```

```
(%o48) [ 16.0, -35.0, 191.0, -1085.0, 6491.0, -3.8885 104, 2.3329 105,
-1.3997 106, 8.3981 106, -5.0388 107, 3.0233 108, -1.814 109, 1.0884
1010, -6.5303 1010, 3.9182 1011, -2.3509 1012, 1.4106 1013, -8.4633
1013, 5.078 1014, -3.0468 1015, 1.8281 1016, -1.0968 1017, 6.5811
1017, -3.9487 1018, 2.3692 1019, -1.4215 1020, 8.5291 1020, -5.1175
1021, 3.0705 1022, -1.8423 1023, 1.1054 1024 ]
```

### 8.3 Prob. 18.20: Repeated Real Roots: $b_1^2 = 4*b_2$

Find the definite solution of

$y[t] - 10*y[t-1] + 25*y[t-2] = 8$ , with  $y[0] = 1$ ,  $y[1] = 5$ .

Comment on stability.

$b_1 = -10$ ,  $b_2 = 25$ ,  $a = 8$ ,  $y_0 = 1$ ,  $y_1 = 5$ .

We take the fast path using the do loop code `ytolist(b1,b2,a,y0,y1,tmax)` defined in `Econ2.mac`.

```
(%i51) tmax : 5$
      tL : makelist (t, t, 0, tmax);
      yL : ylist (-10, 25, 8, 1, 5, tmax);
```


```
(tL) [0,1,2,3,4,5]
```

$$ye = \frac{1}{2}$$

```
(yL) [1,5,33,213,1313,7813]
```

```
(%i52) wxdraw2d (xlabel = "t", ylabel = "y[t]", xrange = [0,tmax],yrange = [-10,1.5e3],
      title = "b1 = -10, b2 = 25, a = 8, y[0] = 1, y[1] = 5", key_pos = top_left,
      points (tL, yL), line_width = 1, color = red, key = "ye = 0.5", explicit(ye, t, 0, tmax))$
```

```
(%t52)
```



A look at longer times without a plot:

```
(%i53) float (ylist (-10, 25, 8, 1, 5, 15) );
```

$$ye = \frac{1}{2}$$

```
(%o53) [1.0,5.0,33.0,213.0,1313.0,7813.0,4.5313 10^4,2.5781 10^5,
      1.4453 10^6,8.0078 10^6,4.3945 10^7,2.3926 10^8,1.2939 10^9,6.958 10^9,
      3.7231 10^10,1.9836 10^11]
```

## 8.4 Prob. 18.34: Complex Roots: $b_1^2 < 4*b_2$

Find the definite solution of

$y[t] + 4*y[t-2] = 15$ , with  $y[0] = 12$ ,  $y[1] = 11$ .


Comment on stability.

For this problem,  $b_1 = 0$ ,  $b_2 = 4$ ,  $a = 15$ . The characteristic equation is  $r^2 + 4 = 0$ . So  $r = \pm \sqrt{-4} = \pm (-4)^{1/2} = \pm i(4)^{1/2} = \pm 2i$ .

Stability for this complex root model depends on the magnitude of  $k = \sqrt{b_2} = \sqrt{4} = 2 > 1$  so unstable.

```
(%i56) tmax : 7$
      tL : makelist (t, t, 0, tmax);
      yL : ylist ( 0, 4, 15, 12, 11, tmax);
(tL)   [0,1,2,3,4,5,6,7]
      ye=3
(ytL)  [12,11,-33,-29,147,131,-573,-509]

(%i57) wxdraw2d (xlabel = "t", ylabel = "y[t]", xrange = [0,tmax],yrange = [-700, 600],
      title = "b1 = 0, b2 = 4, a = 15, y[0] = 12, y[1] = 11",
      points (tL, yL), line_width = 1, color = red, key = "ye = 3", explicit(ye, t, 0, tmax))$
```



(%t57)

Let's look at longer times.

```
(%i58) float (ylist ( 0, 4, 15, 12, 11, 15));
      ye=3
(%o58) [12.0,11.0,-33.0,-29.0,147.0,131.0,-573.0,-509.0,2307.0,
      2051.0,-9213.0,-8189.0,3.6867 104,3.2771 104,-1.4745 105,-
      1.3107 105]
```

We see continued oscillation between positive and negative values, with the steps increasing in size. We see oscillating explosive instability.

## 9 Maxima Solutions Using solve\_rec (eqn, var, [init])

The contributed Maxima package solve\_rec.mac can be used to solve recurrence relation equations.

```
(%i59) load (solve_rec);
```

```
(%o59) C:/maxima-5.43.2/share/maxima/5.43.2/share/solve_rec/solve_rec.mac
```

### 9.1 General Case

```
(%i61) kill (b1, b2, a)$
```

```
gsoln : solve_rec (y[t] + b1*y[t-1] + b2*y[t-2] = a, y[t]);
```

```
(gsoln) 
$$y_t = \frac{\%k_2 (\sqrt{b1^2 - 4 b2} - b1)^t}{2^t} + \frac{\%k_1 (-\sqrt{b1^2 - 4 b2} - b1)^t}{2^t} + \frac{a}{b2 + b1 + 1}$$

```

### 9.2 Numerical Examples

```
(%i62) ysoln (-10, 16, 14, t);
```

```
(%o62) %A1 8^t + %A2 2^t + 2
```

```
(%i63) gs1 : rhs (at (gsoln, [b1 = -10, b2 = 16, a = 14]));
```

```
(gs1) 
$$\frac{\%k_2 16^t}{2^t} + \frac{\%k_1 4^t}{2^t} + 2$$

```

With this version of Maxima, we have to combine the numerator and denominator  $A^t/B^t = (A/B)^t$  by hand in the first two terms to get factors  $8^t$  and  $2^t$ .

Here is a second numerical example:

```
(%i64) ysoln (-2, 1, 8, t);
```

```
(%o64) 4 t^2 + %A2 t + %A1
```

If you try: `gs2 : rhs (at (gsoln, [b1 = -2, b2 = 1, a = 8]));`

you will get a "divide by zero" error message: undefined: 0 to a negative exponent.

You need to start again with solve\_rec (eqn,var) from scratch.

```
(%i65) solve_rec (y[t] + -2*y[t-1] + y[t-2] = 8, y[t]);
```

```
(%o65) 
$$y_t = 4 t^2 + \%k_2 t + \%k_1$$

```

Redoing Example 8 above, with initial conditions,

```
(%i66) solve_rec (y[t] -10*y[t-1] + 16*y[t-2] = 14, y[t], y[0] = 10, y[1] = 36);
```

```
(%o66) y_t=3 8^t+5 2^t+2
```

```
(%i67) grind(%)$
```

```
y[t] = 3*8^t+5*2^t+2$
```

Which agrees with our expression found above for  $y_{def}$  in Example 8.

## 10 **Multiplier-Accelerator model: $a = \text{acceleration}$ , $b = \text{propensity to consume (MPC)}$ , $I_a G_0 = I_a + G_0$**

The multiplier–accelerator model (also known as Keynes–Hansen–Samuelson model) is a macroeconomic model which analyzes the business cycle. This model was developed by Paul Samuelson, who credited Alvin Hansen for the inspiration. This model is based on the Keynesian multiplier, which is a consequence of assuming that consumption intentions depend on the level of economic activity, and the accelerator theory of investment, which assumes that investment intentions depend on the pace of growth in economic activity.

One version of the multiplier–accelerator model follows:

First, the market-clearing level of economic activity is defined as that at which production exactly matches the total of government spending intentions, households' consumption intentions and firms' investing intentions (see below):

$$Y[t] = G[t] + C[t] + I[t],$$

in which investment is the sum of autonomous investment  $I_a$  (a constant) and induced investment  $I_d[t]$ ;

then an equation to express the idea that households' consumption intentions depend upon some measure of economic activity, possibly with a lag ( $b$  is the marginal propensity to consume (MPC),  $(0 < b < 1)$ ):

$$C[t] = b Y[t-1] = \text{consumption expenditure in year } t;$$

then an equation that makes firms' investment intentions react to the pace of change of economic activity (' $a$ ' stands for the accelerator, the "acceleration coefficient", with  $a > 0$ ):

$I_d[t] = a (C[t] - C[t-1]) = a (b Y[t-1] - b Y[t-2]) = a*b (Y[t-1] - Y[t-2]) = \text{induced private firm investment in plant and capital in year } t;$

and finally a statement that government spending intentions are not influenced by any of the other variables in the model. We will take  $G[t] = G_0$ , and combine the constant government spending with the autonomous investment (by firms)  $I_a$ , calling the sum  $I_a G_0$ ,  
 $I_a G_0 = I_a + G_0$ .

We could take  $I_a G_0 = 1$ , and then all other numbers are multiples of  $I_a G_0$ .

We can rearrange these equations and rewrite them as a second-order linear difference equation in the standard form we have been using.

$$Y[t] - b(1+a)Y[t-1] + a*b*Y[t-2] = I_a G_0,$$

in which our constants  $b_1 = -b(1+a)$ ,  $b_2 = a*b$ , and " $a$ " =  $I_a G_0$ .

Based on the signs of  $a$  and  $b$ :  $0 < b < 1$ ,  $a > 0$ ,  $(1+a) > 1$ ,  $b(1+a) > b > 0$ ,  
 $b_1 = -b(1+a) < 0$ , and  $b_2 = a*b > 0$ .

We have two distinct real roots of the characteristic equation  $r^2 - b(1+a)r + a*b = 0$   
 if  $b_1^2 > 4*b_2$ , which here translates to  $b > 4*a/(1+a)^2$ .

We have a double real root if  $b_1^2 = 4*b_2$ , which translates to  $b = 4*a/(1+a)^2$ .

We have complex roots if  $b_1^2 < 4*b_2$  (possible here since  $b_2 > 0$ ), so if  $b < 4*a/(1+a)^2$ .  
 In this complex root case,  $k = b_2^{(1/2)} = \sqrt{a*b} > 0$ , and  
 we need  $|k| < 1$  for convergence.

Let  $y_p$  be the particular solution of the complete equation.

```
(%i69) kill(a,b)$
      yp : ypart (- b*(1+a), a*b, I_a G_0, t);
```

```
(yp)  
$$\frac{I_a G_0}{-(a+1)b + a b + 1}$$

```

The denominator simplifies if you expand it out.

```
(%i70) yp : num(yp)/expand(denom(yp));
```

```
(yp)  
$$\frac{I_a G_0}{1-b}$$

```

The expression  $1/(1-b)$  is the multiplier that would multiply  $I_a G_0$  the absence of induced investment. The quantity  $I_a G_0/(1-b)$  is the "intertemporal equilibrium income"  $Y_e$ .

## 10.1 yMA (a, b, $I_a G_0$ , t) Multiplier-Accel. Indefinite Solution



We can piggy-back on our previous work and define a solution for  $Y[t]$  using a new Maxima function  $yMA(a, b, laG0, t)$ , since we know how to relate  $b_1$  and  $b_2$  to the model parameters  $a$  and  $b$ . Recall  $a$  is the acceleration and  $b$  is the propensity to consume and  $b_1 = -b(1+a) < 0$ ,  $b_2 = a*b > 0$ .

```
(%i71) yMA (aa, bb,iag0, tt) := ysoln (-bb*(1 + aa), aa*bb, iag0, tt)$
```

### 10.1.1 yMA(a, b, laG0, t) Method Example

As an example, we assume  $a = 0.5$ ,  $b = 0.9$ ,  $laG0 = 1$ ,  $Y[0] = 1$ ,  $Y[1] = 2$ .

The first step is to use  $yMA$  to calculate the indefinite solution; that solution includes two arbitrary constants  $\%A1$  and  $\%A2$ .

```
(%i72) yindef : yMA (0.5, 0.9, 1, t);
```

```
(yindef) %A1 0.75t + %A2 0.6t + 10.0
```

To define a Maxima function  $ys(t)$ , use two single quotes (') in front of  $yindef$ :

```
(%i73) ys(t) := "yindef;
```

```
(%o73) ys (t) := %A1 0.75t + %A2 0.6t + 10.0
```

```
(%i74) solns : solve ([ys(0) = 1, ys(1) = 2]), numer;
```

```
(solns) [[ %A2=8.3333, %A1=-17.3333]]
```

Now define the expression  $ydef$  (the definite solution, taking into account the two known values of  $Y[t]$ ).

```
(%i75) ydef : at (yindef, solns[1]);
```

```
(ydef) -17.333 0.75t + 8.3333 0.6t + 10.0
```

To easily generate a list of values of  $Y[t]$ , define the Maxima function  $yd(t)$ , again using two single quotes (') in front of the expression  $ydef$ :

```
(%i76) yd(t) := "ydef;
```

```
(%o76) yd (t) := -17.333 0.75t + 8.3333 0.6t + 10.0
```

```
(%i78) tmax : 10$
```

```
tL : makelist (t, t, 0, tmax);
```

```
(tL) [0,1,2,3,4,5,6,7,8,9,10]
```

```
(%i79) ytL : map ('yd, tL);
```

```
(ytL) [1.0,2.0,3.25,4.4875,5.5956,6.5347,7.3038,7.9196,8.4047,8.7825,
9.0743]
```

An alternative route to a list of values of  $Y[t]$  is the following:


```
(%i80) makelist (subst (tL[j], t, ydef), j, 1, length (tL));
```

```
(%o80) [1.0,2.0,3.25,4.4875,5.5956,6.5347,7.3038,7.9196,8.4047,8.7825,
9.0743]
```

We can then make a plot of the values of  $Y[t]$  over the chosen interval of times. We make use of the draw2d function points (xL,yL).

```
(%i81) wxdraw2d (xlabel = "t", ylabel = "Y[t]", xrange = [0,tmax],yrange = [0, 12],
title = " a = 0.5, b = 0.9, IaG0 = 1, Y[0] = 1, Y[1] = 2",
key_pos = bottom_right, points (tL, ytL), line_width = 1, color = red,
key = "Ye = 10", explicit(10, t, 0, tmax))$
```

```
(%t81)
```



We can map yd onto any list of times to see later behavior: for example from  $t = 10$  to  $t = 20$ :

```
(%i82) map ('yd, makelist (t, t, 10, 20));
```

```
(%o82) [9.0743,9.2982,9.4691,9.5991,9.6977,9.7723,9.8286,9.8711,
9.9031,9.9272,9.9453]
```

### 10.1.2 MAlist(a,b,IaG0,Y0,Y1,tmax), makeLists(L) Method Example

To examine details (consumption  $C[t]$  and induced investment  $Id[t]$ ), we use the function `MAlist` ( $a$ ,  $b$ ,  $laG0$ ,  $Y[0]$ ,  $Y[1]$ ,  $tmax$ ), defined in `Econ2.mac`, which uses Maxima do loop code. This Maxima function creates a list of lists which can then be used to make a table (using `matrix`) and also a plot of  $C[t]$  and  $Id[t]$  as well as  $Y[t]$ . Indeed, this do loop code is a faster route to see what is happening to just  $Y[t]$  itself. The function `MAlist` also produces a global value for  $Ye = laG0/(1-b)$ .

As an example, we choose  $a = 0.5 = 1/2$ ,  $b = 0.9$ ,  $laG0 = 1$ , then  $Ye = 1/(1-b) = 10$ . We assume  $Y[0] = 1$ ,  $Y[1] = 2$ , and then use:

$Y[t] = C[t] + Id[t] + laG0 = b*Y[t-1] + a*b*(Y[t-1] - Y[t-2]) + laG0$ ,  
to advance the solution to the next integral time.

Since we are going to do two things with the output of `MAlist`, we assign the output (a list of lists) to the name `maL`. The sublists start with a list of titles for each column, followed by lists of numbers holding succeeding values  $[t, C[t], Id[t], laG0[t], Y[t]]$ , in which  $laG0$  is the sum  $la + G0$  (autonomous investment + government spending). The value of the "intertemporal equilibrium"  $Ye$  is calculated and printed to the screen.

```
(%i83) maL : MAlist (0.5, 0.9, 1, 1, 2, 11);
```

```
Ye=10.0
```

```
(maL) [[t, Ct, Idt, la + G0, Yt], [0, 0, 0, 1, 1], [1, 0.9, 0, 1, 2], [2, 1.8,
0.45, 1, 3.25], [3, 2.925, 0.5625, 1, 4.4875], [4, 4.0388, 0.55688, 1,
5.5956], [5, 5.0361, 0.49866, 1, 6.5347], [6, 5.8812, 0.42259, 1, 7.3038
], [7, 6.5735, 0.3461, 1, 7.9196], [8, 7.1276, 0.27707, 1, 8.4047], [9,
7.5642, 0.2183, 1, 8.7825], [10, 7.9043, 0.17003, 1, 9.0743], [11,
8.1669, 0.1313, 1, 9.2982]]
```

Here is how to use the `matrix` function with a list of lists to make a table, letting the `matrix` function do all the work of spacing the columns so the entries can be easily understood.

```
(%i84) apply ('matrix, maL);
```

$t$	$Ct$	$Idt$	$la + G0$	$Yt$
0	0	0	1	1
1	0.9	0	1	2
2	1.8	0.45	1	3.25
3	2.925	0.5625	1	4.4875
4	4.0388	0.55688	1	5.5956
5	5.0361	0.49866	1	6.5347
6	5.8812	0.42259	1	7.3038
7	6.5735	0.3461	1	7.9196
8	7.1276	0.27707	1	8.4047
9	7.5642	0.2183	1	8.7825
10	7.9043	0.17003	1	9.0743
11	8.1669	0.1313	1	9.2982

```
(%o84)
```

Inside MAlist, the first row is a list of column titles, which is entered using double quotes ("") to create Maxima strings, in the form

```
["t", "Ct", "Idt", "la + G0", "Yt" ].
```

A second function defined in Econ2.mac is

```
makeLists(L).
```

This is a Maxima function which turns the list produced by MAlist into globally known

sublists which are defined as (globally known names): tL (a list of times),

CtL (a list of values of  $C[t]$ ), IdtL (a list of values of  $Id[t]$ ), laG0L (a list of values of  $laG0$ ),

and finally YtL (a list of values of  $Y[t]$ ).

```
(%i85) makeLists (maL);
```

```
(%o85) done
```

The list names are now globally known and we can separately examine each of these lists.

```
(%i86) tL;
```

```
(%o86) [0,1,2,3,4,5,6,7,8,9,10,11]
```

```
(%i87) CtL;
```

```
(%o87) [0,0.9,1.8,2.925,4.0388,5.0361,5.8812,6.5735,7.1276,7.5642,7.9043,8.1669]
```

(%i88) IdtL;

(%o88) [0,0,0.45,0.5625,0.55688,0.49866,0.42259,0.3461,0.27707,  
0.2183,0.17003,0.1313]

(%i89) IaG0L;

(%o89) [1,1,1,1,1,1,1,1,1,1,1,1,1]

(%i90) YtL;


(%o90) [1,2,3.25,4.4875,5.5956,6.5347,7.3038,7.9196,8.4047,8.7825,  
9.0743,9.2982]

We now make a plot, ignoring the list IaG0L, to show values of Y[t], C[t], and Id[t] vs. time t.

(%i92) tmax : 11\$


```
wxdraw2d(xlabel = "t", xrange = [0,tmax], yrange = [-3,16], key_pos = top_left,
title = "a = 0.5, b = 0.9, IaG0 = 1, Y[0] = 1, Y[1] = 2",
point_type = filled_circle, key = "Yt",points(tL,YtL),color = red,key = "Ct",
points (tL, CtL), color = brown,key = "Idt",points(tL,IdtL),
color = magenta, line_width = 1,key = sconcat( "Ye = ",Ye), explicit(Ye, t, 0,tmax),
color = black, key = "", explicit (0, t, 0, tmax))$
```

(%t92)



Of course we can just plot Y[t] vs. t, if we are not interested in details.

```
(%i94) tmax : 11$
wxdraw2d (xlabel = "t", ylabel = "Y[t]", xrange = [0, tmax], yrange = [0, 16],
title = " a = 0.5, b = 0.9, IaG0 = 1, Y[0]=1, Y[1]=2",
points (tL, YtL), line_width = 1, color = red,
key = sconcat( "Ye = ", Ye), explicit(Ye, t, 0, tmax))$
```




Here is a brief example which ignores details but looks at  $Y[t]$  behavior for longer times. We are continuing to use the same model parameters and initial conditions, but increasing  $tmax$  from 11 to 20.

```
(%i97) MAlist (0.5, 0.9, 1, 1, 2, 20)$
makeLists (%)$
YtL;
Ye=10.0
(%o97) [1,2,3.25,4.4875,5.5956,6.5347,7.3038,7.9196,8.4047,
8.7825,9.0743,9.2982,9.4691,9.5991,9.6977,9.7723,9.8286,9.8711,
9.9031,9.9272,9.9453]
```

Let's check  $tL$  also:

```
(%i98) tL;
(%o98) [0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20]
```

```
(%i100) tmax : 20$
wxdraw2d (xlabel = "t", ylabel = "Y[t]", xrange = [0, tmax], yrange = [0, 16],
title = " a = 0.5, b = 0.9, IaG0 = 1, Y[0]=1, Y[1]=2",
points (tL, YtL), line_width = 1, color = red,
key = sconcat( "Ye = ", Ye), explicit(Ye, t, 0, tmax))$
```



(%t100)

## 10.2 Properties of the Characteristic Roots

If  $r_1$  and  $r_2$  are the characteristic roots,  $r_1$  and  $r_2$  are always related to each other by the following three relations: (our definitions of  $r_1$  and  $r_2$  agree with Chiang and Wainwright (they use  $(b_1, b_2)$  as symbols for the characteristic roots, we use  $(r_1, r_2)$ ).

- 1.)  $r_1 + r_2 = b(1 + a) > 0,$
- 2.)  $r_1 r_2 = a b > 0,$
- 3.)  $0 < (1 - r_1)(1 - r_2) < 1.$

Here we derive these properties using Maxima.

```
(%i102) kill (r,a,b)$
solns : solve (r^2 - b*(1+a)*r + a*b = 0, r);
(solns) [r = - (sqrt((a^2+2 a+1) b^2-4 a b+(-a-1) b) / 2), r =
sqrt((a^2+2 a+1) b^2-4 a b+(a+1) b) / 2]
```

```
(%i103) [r2, r1] : map ('rhs, solns);
```

```
(%o103) [ -  $\frac{\sqrt{(a^2+2a+1)b^2-4ab+(-a-1)b}}{2}$ ,  

 $\frac{\sqrt{(a^2+2a+1)b^2-4ab+(a+1)b}}{2}$  ]
```

The numerator and denominator of r2 are:

```
(%i104) num(r2);
```

```
(%o104)  $-\sqrt{(a^2+2a+1)b^2-4ab}-(-a-1)b$ 
```

```
(%i105) denom(r2);
```

```
(%o105) 2
```

It should be clear from this definition that  $r_2 < r_1$ , (provided  $r_1$  and  $r_2$  are distinct real numbers) since for  $r_2$  we are subtracting the square root quantity from the positive number  $b*(1+a)$ , whereas for  $r_1$  we are adding.

```
(%i106) rsum : r1 + r2, expand, factor;
```

```
(rsum)  $(a+1)b$ 
```

```
(%i107) rprod : expand (r1*r2);
```

```
(rprod)  $ab$ 
```

```
(%i108) grind(rprod)$
```

```
 $a*b$ 
```

```
(%i109) rprod2 : (1 - r1)*(1 - r2), expand;
```

```
(rprod2)  $1-b$ 
```

Since  $0 < b < 1$ ,

$(1-b) > 0$  and  $(1-b) < 1$ .

Hence  $0 < (1 - r_1)*(1 - r_2) < 1$ .

### 10.3 Case 1: Two Distinct Real Roots: $b > 4*a/(1+a)^2$

Assuming  $r_1$  and  $r_2$  are two distinct real roots, relation (2) above implies that neither root can be equal to zero, and both  $r_1$  and  $r_2$  must have the same sign. Then relation (1) above implies that  $r_1$  and  $r_2$  cannot both be negative, hence  $r_1$  and  $r_2$  must both be positive:  $r_1 > 0$ ,  $r_2 > 0$ . Hence there can be no oscillation in the time path.




### 10.3.1 Case 1A: $a < 1$ , $0 < r_2 < r_1 < 1$ , convergence

For Case 1A,  $r_1$  and  $r_2$  are both real positive numbers less than one,  $0 < r_2 < r_1 < 1$ . This occurs if  $0 < a < 1$  and  $b > 4*a/(1+a)^2$ .

Let's make a plot which paints the area of the  $(a,b)$  plane for this real root convergence case in dark gray.

```
(%i111) [xmax,yxmax] : [5,3]$
wxdraw2d(xlabel = "a",ylabel = "b",xrange = [0,xmax], yrange = [0,yxmax],
  key = "1/a", explicit (1/a,a,0.3,xmax), key = "real root convergence",
  filled_func = true, fill_color = dark_gray, explicit(1,a,0,1), color = red,
  fill_color = light_gray, key = "", explicit(4*a/(1+a)^2, a, 0,1),
  key = "4*a/(1+a)^2",filled_func = false, explicit(4*a/(1+a)^2, a, 0,xmax),
  color = black, line_width = 1,key = "",explicit (1,a,0,xmax), parametric(1,yy,yy,0,1))$
```



We can numerically survey values of the roots  $r_1$  and  $r_2$  for values of  $b$  belonging to that part of the range  $0 < b < 1$  for which we get distinct real roots, for given values of  $a$ , using two functions:  $bmin(a)$  and  $survey(a)$  defined here.

The function  $survey(a)$  constructs a list of  $b$  values which respect  $b > 4*a/(1+a)^2$  (so we are dealing with distinct real roots), and then evaluates  $r_1$  and  $r_2$  at each  $(a,b)$  point.

It is easier to type  $1e-10$  than  $10^{(-10)}$ , and they produce the same result in Maxima.

```
(%i112) is(equal (1e-10, 10^(-10) ));
```

```
(%o112) true
```

```
(%i113) bmin(x) := 4*x/(1+x)^2$
```

```
(%i114) survey(%a) := block ([blist, bstart,bend,delb,xx,yy,ss,%r1,%r2],
    bstart : bmin (%a) + 1e-10,
    bend : 1 - 1e-10,
    delb : (bend - bstart)/10,
    blist : makelist (bstart + n*delb,n, 0,10),
    print (" [b, r1 , r2]"),
    for %b in blist do (
        xx : %b*(1+%a),
        yy : xx^2 - 4*%a*%b,
        ss : sqrt(yy),
        %r1 : (xx+ss)/2,
        %r2 : (xx - ss)/2,
        print ([%b, %r1, %r2])),
    done)$
```

For example, for  $a = 0.5$  (we increase the number of digits printed out):

```
(%i116) fpprintprec : 16$
survey (0.5)$
[b, r1 , r2]
[0.8888888888888888,0.666673737809771,0.6666595956735623]
[0.9000000000799999,0.7500000003333331,0.5999999997866667]
[0.91111111111711111,0.7900520708164651,0.5766145959402016]
[0.9222222222622222,0.8231644487622463,0.560168884631087]
[0.9333333333533333,0.8527525232162003,0.5472474768137996]
[0.9444444444444444,0.8801294010674023,0.5365372655992642]
[0.9555555555355555,0.9059636114812999,0.5273697218220332]
[0.9666666666266667,0.9306493779004148,0.5193506220395852]
[0.9777777777177777,0.9544416525639517,0.5122250140127148]
[0.9888888888088889,0.977516194804202,0.5058171384091313]
[0.9999999999,0.999999998,0.50000000005]
```

We first use the more long-winded multiplier-acceleration model function  $yMA(a,b,laG0,t)$  method to find the indefinite solution for  $Y[t]$ , taking  $a = 0.5$  and  $b = 0.9$ ,  $laG0 = 1$ , and then the definite solution assuming  $Y[0] = 20$ , and  $Y[1] = 18$ . (We then use the shorter path using  $MAlist$  and  $makeLists$ ).

```
(%i117) bmin (0.5);
```

```
(%o117) 0.8888888888888888
```

```
(%i119) fpprintprec : 5$
```

```
yindef : yMA (0.5, 0.9, 1, t);
```

```
(yindef) %A1 0.75t + %A2 0.6t + 10.0
```

Use two single quotes (') in front of yindef to define a Maxima function ys(t).

```
(%i120) ys(t) := "yindef;
```

```
(%o120) ys(t) := %A1 0.75^t + %A2 0.6^t + 10.0
```

We take Y[0] to be equal to 20, and Y[1] = 18, and solve for the adjustable constants.

```
(%i121) solns : solve ([ys(0) = 20, ys(1) = 18]), numer;
```

```
(solns) [[%A2 = -3.3333, %A1 = 13.333]]
```

```
(%i122) ydef : at (yindef, solns [1]);
```

```
(ydef) 13.333 0.75^t - 3.3333 0.6^t + 10.0
```

```
(%i123) yd(t) := "ydef;
```

```
(%o123) yd(t) := 13.333 0.75^t - 3.3333 0.6^t + 10.0
```

```
(%i125) tL : makelist (t, t, 0, 11);
```

```
YtL : map ('yd, tL);
```

```
(tL) [0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11]
```

```
(YtL) [20.0, 18.0, 16.3, 14.905, 13.787, 12.905, 12.218, 11.686, 11.279,  
10.968, 10.731, 10.551]
```


```
(%i127) tmax : 11$
```

```
wxdraw2d (xlabel = "t", ylabel = "Y[t]", xrange = [0, tmax], yrange = [0, 30],
```

```
title = " discrete real roots, a = 0.5, b = 0.9, IaG0 = 1, Y[0] = 20, Y[1] = 18",
```

```
points (tL, YtL), line_width = 1, color = red, key = "Ye = 10", explicit(10, t, 0, tmax))$
```

```
(%t127)
```




In this case in which  $a < 1$ , and  $b > 4*a/(1+a)^2$ , we see steady progression to the equilibrium value  $Y_e = 10$ , sign of a stable solution.

It is quicker to use our do-loop code `MAlist (a,b,IaG0,Y0,Y1, tmax)` defined in `Econ2.mac`.

```
(%i130) MAlist (0.5, 0.9, 1, 20, 18, 11)$
makeLists(%)$
YtL;
Ye=10.0
(%o130) [[20, 18, 16.3, 14.905, 13.787, 12.905, 12.218, 11.686, 11.279,
10.968, 10.731, 10.551]
```

```
(%i132) tmax : 11$
wxdraw2d (xlabel = "t", ylabel = "Y[t]", xrange = [0, tmax], yrange = [0,30],
title = "discrete real roots, a = 0.5, b = 0.9, IaG0 = 1, Y[0] = 20, Y[1] = 18",
points (tL, YtL), line_width = 1, color = red,
key = sconcat("Ye = ", Ye), explicit(Ye, t, 0, tmax))$
```




For the details of consumption and induced investment, we use the already defined lists `CtL`, `IdtL`.

```
(%i133) [CtL, IdtL];
(%o133) [[0, 18.0, 16.2, 14.67, 13.414, 12.408, 11.614, 10.996, 10.518, 10.151,
9.8708, 9.6576], [0, 0, -0.9, -0.765, -0.62775, -0.50321, -0.39685, -0.3093,
-0.23897, -0.18343, -0.14009, -0.10658]]
```

(%i135) tmax : 11\$

```
wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [-3,30],
  point_type = filled_circle, key = "Yt", points(tL, YtL), color = red, key = "Ct",
  points(tL, CtL), color = brown, key = "Idt", points(tL, IdtL),
  color = magenta, line_width = 1, key = "Ye = 10", explicit(Ye, t, 0, tmax),
  color = black, key = "", explicit(0, t, 0, tmax))$
```



Nothing dramatic here.


### 10.3.2 Case 1B: $a > 1$ , $r_1 > r_2 > 1$ , divergence

Case 1B assumes  $r_2$  and  $r_1$  are both numbers greater than 1.  $r_1 > r_2 > 1$ .

We satisfy relation (3) provided  $(1 - r_1)(1 - r_2)$  [a product of two negative numbers and hence positive] is less than 1, so we need  $(1 - r_1)(1 - r_2) < 1$ . But we have shown above that with  $0 < b < 1$ ,  $(1 - r_1)(1 - r_2) = 1 - b < 1$ . Relation (2) becomes  $a*b = r_1*r_2 > 1$  since both  $r_1$  and  $r_2$  are greater than 1.

Let's make a plot which paints the real root divergence region of the  $(a,b)$  plane in dark gray.

```
(%i137) [xmax,ymax] : [5, 3]$
wxdraw2d(xlabel = "a", ylabel = "b", xrange = [0, xmax], yrange = [0, ymax],
filled_func = true, fill_color = dark_gray, key = "real root divergence",
explicit (1, a, 1, xmax), key = "", fill_color = light_gray,
explicit (4*a/(1+a)^2, a, 1, xmax), filled_func = false,
key = "4*a/(1+a)^2", explicit (4*a/(1+a)^2, a, 0, xmax), color = red, key = "1/a",
explicit (1/a, a, 0.3, xmax), color = black, line_width = 1, key = "",
explicit (1,a, 0, xmax), parametric (1, yy, yy, 0, 1))$
```



As an example, we consider  $a = 1.5$ .

The value of  $b$  must then lie in the range:  $b_{\min}(1.5) < b < 1$ .

```
(%i138) bmin(1.5);
```

```
(%o138) 0.96
```

```
(%i140) fpprintprec : 16$
survey (1.5)$
 [b, r1, r2]
 [0.9600000001, 1.200012247574221, 1.199987752675779]
 [0.96400000008, 1.282620874360731, 1.127379125839269]
 [0.96800000006, 1.320000000490909, 1.099999999659091]
 [0.97200000004, 1.350000000277776, 1.079999999822223]
 [0.97600000002, 1.376204993642362, 1.063795006407638]
 [0.98, 1.400000000000001, 1.049999999999999]
 [0.98399999998, 1.422093727015994, 1.037906272934006]
 [0.98799999996, 1.442906228660204, 1.027093771239796]
 [0.99199999994, 1.462710574222674, 1.017289425627325]
 [0.99599999992, 1.481696007198392, 1.008303992601608]
 [0.9999999999, 1.4999999995, 1.0000000002]
```

As an example of this divergent region of the (a,b) plane, we choose  $a = 1.5$  and  $b = 0.97$ ,  $laG0 = 1$ ,  $Y[0] = 10$ ,  $Y[1] = 15$ .


```
(%i144) fpprintprec : 5$
MAlist (1.5, 0.97, 1, 10, 15, 7)$
makeLists(%)$
YtL;
Ye=33.333
(%o144) [10, 15, 22.825, 34.526, 51.514, 75.687, 109.59, 156.63]
```

This already shows an increase in step size each succeeding time interval, with no sign of slowing down near 33.3.

(%i146) tmax : 7\$

```
wxdraw2d (xlabel = "t", ylabel = "Y[t]", xrange = [0, tmax], yrange = [0,200],
  title = " a = 1.5, b = 0.97, IaG0 = 1, Y[0] = 10, Y[1] = 15", key_pos = top_left,
  points (tL, YtL), line_width = 1,
  color = red, key = sconcat( "Ye = ", Ye), explicit(Ye, t, 0, tmax))$
```

(%t146)




In this example in which  $a > 1$ , and  $b > 4*a/(1+a)^2$ , we see larger jumps each succeeding time interval, blowing past the intertemporal equilibrium value  $Ye = 33.3$ , an example of "explosive instability".

For the details:



```
(%i148) tmax : 7$
wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [0, 250], key_pos = top_left,
title = "a = 1.5, b = 0.97, IaG0 = 1, Y[0] = 10, Y[1] = 15",
point_type = filled_circle, key = "Yt",points(tL,YtL),color = red,key = "Ct",
points (tL, CtL), color = brown,key = "Idt",points(tL,IdtL),
color = magenta, line_width = 1,key = sconcat ( "Ye = ", Ye), explicit(Ye, t, 0, tmax))$
```

(%t148)




This unregulated economy is going gangbusters, with both consumption and induced investment rising steadily, after we start with a rising GDP  $Y[t]$ .

But with the same acceleration and propensity to consume parameters (a,b), if we start with a large but declining GDP, it's all downhill.

```
(%i151) MAlist (1.5, 0.97, 1, 54, 50, 7)$
makeLists (%)$
YtL;
Ye=33.333
(%o151) [54,50,43.68,34.174,20.318,0.54689,-27.236,-65.843]
```

```
(%i153) tmax : 7$
```

```
wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [-100, 150],
title = "a = 1.5, b = 0.97, IaG0 = 1, Y[0] = 54, Y[1] = 50",
point_type = filled_circle, key = "Yt", points(tL, YtL), color = red, key = "Ct",
points (tL, CtL), color = brown, key = "Idt", points(tL, IdtL),
color = magenta, line_width = 1, key = sconcat ( "Ye = ", Ye), explicit(Ye, t, 0, tmax),
color = black, key = "", explicit (0, t, 0, tmax))$
```



## 10.4 Case 2: Double Roots: $b = 4*a/(1+a)^2$ , $rval(a)$ , $ydouble(a, IaG0, t)$

This is the case in which  $b^2 = 4*a$ , or  $b = 4*a/(1+a)^2$ . From our general relations for  $(r_1, r_2)$ , the relation  $r_1 + r_2 = b*(1+a)$ , and using  $r = r_1 = r_2$ , we get  $r = b*(1+a)/2$ .

Because  $a > 0$  and  $0 < b < 1$ , we have the single root  $r > 0$ , so no oscillation.

Replacing  $b$  with  $4*a/(1+a)^2$ , we get:

$$r = 2*a/(1+a).$$

If  $r < 1$  then  $2*a < 1 + a$ , and  $a < 1$ .

Likewise if  $r > 1$  then  $2*a > 1 + a$ , and  $a > 1$ .

```
(%i154) rval(_a) := 2*_a/(1 + _a)$
```

```
(%i155) [rval(0.5), rval(1.5)];
```

```
(%o155) [0.66667, 1.2]
```

We can define  $ydouble(a, IaG0, t)$  in terms of  $yMA(a, b, IaG0, t)$ :

```
(%i156) ydouble(aa,iag0,tt) := yMA(aa, bmin(aa), iag0, tt)$
```

### 10.4.1 Case 2A: $a < 1$ , $r < 1$ , convergence

As an example of a converging solution, we take  $a = 0.5$ , and  $laG0 = 1$ .  
We also take  $Y[0] = 15$ ,  $Y[1] = 13$ .

Built into this case is  $b = bmin(a) = 4*a/(1+a)^2$ .

```
(%i157) bmin(0.5);
```

```
(%o157) 0.88889
```

```
(%i158) yindef : ydouble (0.5, 1, t);
```

```
(yindef) %A2 t 0.66667t + %A1 0.66667t + 9.0
```

```
(%i159) ys(t) := "yindef;
```

```
(%o159) ys(t) := %A2 t 0.66667t + %A1 0.66667t + 9.0
```

```
(%i160) solns : solve ([ys(0) = 15, ys(1) = 13]),numer;
```

```
(solns) [[ %A2=0, %A1=6 ]]
```

```
(%i161) ydef : at (yindef, solns[1]);
```

```
(ydef) 6 0.66667t + 9.0
```

```
(%i162) grind(%)$
```

```
6*0.66667t+9.0$
```

```
(%i163) yd(t) := "ydef;
```

```
(%o163) yd(t) := 6 0.66667t + 9.0
```

```
(%i165) tmax : 10$
```


```
tL : makelist (t, t, 0, tmax);
```

```
(tL) [0,1,2,3,4,5,6,7,8,9,10]
```

```
(%i166) YtL : map ('yd, tL);
```

```
(YtL) [15.0,13.0,11.667,10.778,10.185,9.7901,9.5267,9.3512,9.2341,  
9.1561,9.104]
```

```
(%i167) wxdraw2d (xlabel = "t", ylabel = "Y[t]", xrange = [0, tmax], yrange = [0, 20],
  title = " double real root, a = 0.5, b = 4*a/(1+a)^2, IaG0 = 1",
  points (tL, YtL), line_width = 1,
  color = red, key = "Ye = 9", explicit(9, t, 0, tmax))$
```



In this example with  $a = 0.5 < 1$  we see a steady progression towards  $Y_e = 9$ , with each step smaller than the previous step.

Details, starting  $Y[0]$  above  $Y_e$  ( $Y[0] = 15$ ,  $Y[1] = 13$ ):

```
(%i170) MAlist (0.5, bmin(0.5), 1, 15, 13, 10)$
  makeLists (%)$
  YtL;
```


$Y_e = 9.0$

```
(%o170) [ 15, 13, 11.667, 10.778, 10.185, 9.7901, 9.5267, 9.3512, 9.2341,
  9.1561, 9.104 ]
```

(%i172) `tmax : 10`

```
wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [-2, 25],
title = "a = 0.5, b = 4*a/(1+a)^2, IaG0 = 1, Y[0] = 15, Y[1] = 13",
point_type = filled_circle, key = "Yt",points(tL,YtL),color = red,key = "Ct",
points (tL, CtL), color = brown,key = "Idt",points(tL,IdtL),
color = magenta, line_width = 1,key = sconcat ( "Ye = ", Ye), explicit(Ye, t, 0,tmax),
color = black, key = "", explicit (0, t, 0, tmax))$
```

(%t172)



Details, starting Y[0] below Ye:

(%i175) `MAlist (0.5, bmin(0.5), 1, 3, 5, 10)`

`makeLists (%)`

`YtL;`

`Ye=9.0`


(%o175) `[3,5,6.3333,7.2222,7.8148,8.2099,8.4733,8.6488,8.7659,8.8439,8.896]`

(%i177) `tmax : 10`

```

wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [-3, 20], key_pos = top_left,
title = "a = 0.5, b = 4*a/(1+a)^2, IaG0 = 1, Y[0] = 3, Y[1] = 5",
point_type = filled_circle, key = "Yt",points(tL,YtL),color = red,key = "Ct",
points (tL, CtL), color = brown,key = "Idt",points(tL,IdtL),
color = magenta, line_width = 1,key = sconcat( "Ye = ", Ye), explicit(Ye, t, 0, tmax),
color = black, key = "", explicit (0, t, 0, tmax))$
    
```

(%t177)



### 10.4.2 Case 2B: $a > 1, r > 1$ , divergence

(%i180) `MAlist (1.5, bmin(1.5), 1, 10, 13, 10)`

`makeLists (%)`


`YtL;`

`Ye=25.0`

(%o180) `[ 10, 13, 17.8, 25.0, 35.368, 49.883, 69.79, 96.664, 132.5, 179.79, 241.71 ]`

(%i182) `tmax : 10`\$

```
wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [-20, 250], key_pos = top_left,
title = "a = 1.5, b = 4*a/(1+a)^2, IaG0 = 1, Y[0] = 10, Y[1] = 13",
point_type = filled_circle, key = "Yt", points(tL, YtL), color = red, key = "Ct",
points(tL, CtL), color = brown, key = "Idt", points(tL, IdtL),
color = magenta, line_width = 1, key = sconcat("Ye = ", Ye), explicit(Ye, t, 0, tmax),
color = black, key = "", explicit(0, t, 0, tmax))$
```



(%t182)

Details, with a declining GDP at the start:

(%i185) `MAlist (1.5, bmin(1.5), 1, 35, 32, 10)`\$

`makeLists (%)`\$

`YtL;`


`Ye=25.0`

(%o185) `[35,32,27.4,20.68,11.176,-1.9568,-19.79,-43.678,-75.329,-116.89,-171.07]`

(%i187) tmax : 10\$

```
wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [-200, 60], key_pos = bottom_left,
title = "a = 1.5, b = 4*a/(1+a)^2, IaG0 = 1, Y[0] = 35, Y[1] = 32",
point_type = filled_circle, key = "Yt",points(tL,YtL),color = red,key = "Ct",
points (tL, CtL), color = brown,key = "Idt",points(tL,IdtL),
color = magenta, line_width = 1,key = sconcat ( "Ye = ", Ye), explicit(Ye, t, 0, tmax),
color = black, key = "", explicit (0, t, 0, tmax))$
```

(%t187)




### 10.5 Case 3: Complex Roots: $b < 4*a/(1+a)^2$

#### 10.5.1 Case 3A: $b < \min(4*a/(1+a)^2, 1/a)$ , $\sqrt{a*b} < 1$ , convergence

Let's make a plot to show the region in the (a,b) plane which produces this case.



```
(%i188) wxdraw2d(xlabel = "a",ylabel = "b",key = "4*a/(1+a)^2",
    explicit (4*a/(1+a)^2,a,1,5),color = red,key = "1/a",
    explicit (1/a,a,0.5, 1), filled_func = true, fill_color = dark_gray,
    key = "complex roots, k<1", explicit (4*a/(1+a)^2,a,0,1),
    key = "", explicit(1/a,a,1,5),filled_func = false,
    color = blue, explicit(4*a/(1+a)^2,a,0,1),color = red,explicit(1/a,a,1,5),
    color = black,line_width = 1, explicit (1,a,0,5),parametric(1,yy,yy,0,1))$
```



This plot shows that for  $0 < a < 1$ , we need  $0 < b < 4*a/(1+a)^2$  and for  $a > 1$  we need  $0 < b < 1/a$ , to have converging  $k < 1$  solutions arising from complex root cases.

The analytic solution for complex roots has the form  $k^t * (\text{sinusoid})$ , where  $k = \sqrt{a*b}$ , and for convergence we need  $k < 1$ .

For given values of  $a$ , survey the value of  $k = \sqrt{a*b}$  for  
 $0 < b < \min(4*a/(1+a)^2, 1/a)$   
 using `surveyComplex1(a)`:

```
(%i189) surveyComplex1(%a) :=
  block ([blist, bstart,bend,delb],
    bstart : 1e-10,
    if %a <= 0 then print("%a must be greater than zero")
    else (
      if %a < 1 then bend : 4*%a/(1+%a)^2 - 1e-10
      else bend : 1/%a - 1e-10,
      delb : (bend - bstart)/10,
      /* print("bstart = ",bstart,"bend = ", bend," delb = ",delb), */
      blist : makelist (bstart + n*delb,n, 0,10),
      print (" [b, k]"),
      for %b in blist do print ( [%b, sqrt (%a*%b)] ),
    done)$
```

```
(%i191) fpprintprec:16$
  surveyComplex1(0.5);
  [b, k]
  [1.0 10-10, 7.071067811865476 10-6]
  [0.08888888896888887, 0.2108185107727603]
  [0.1777777778377778, 0.2981423970502835]
  [0.2666666667066666, 0.3651483716974968]
  [0.355555555755555, 0.4216370213676424]
  [0.4444444444444444, 0.4714045207910317]
  [0.5333333333133332, 0.5163977794846397]
  [0.6222222221822221, 0.5577733510047885]
  [0.711111111051111, 0.5962847939747881]
  [0.7999999999199999, 0.6324555320020531]
  [0.8888888887888888, 0.666666666291666]
(%o191) done
```

Experimenting with `surveyComplex1(a)` shows that with

$$0 < b < \min(4*a/(1+a)^2, 1/a)$$

leads to  $k < 1$  and a convergent solution, since the complementary solution has the form

$$k^t [A1 \cos(th*t) + A2 \sin(th*t)]$$

with real adjustable constants  $A1$  and  $A2$ .

As a complex root example showing convergence we choose  $a = 0.5$ ,  $b = 0.5$  (see plot above),  $laG0 = 1$ ,  $Y[0] = 6$ ,  $Y[1] = 5$ , in which  $k = \sqrt{a*b} = 1/2 < 1$  and  $Ye = 2$ .

```
(%i193) fpprintprec : 5$
  yindef : yMA(0.5,0.5, 1, t);
(yindef) (%A2 sin(0.72273 t) + %A1 cos(0.72273 t)) 0.5t+2.0
```

```
(%i194) ys(t) := "yundef";
```

```
(%o194) ys(t) := (%A2 sin(0.72273 t) + %A1 cos(0.72273 t)) 0.5^t + 2.0
```

```
(%i195) solns : solve([ys(0) = 6, ys(1) = 5]), numer;
```

```
(solns) [[%A2=4.5356, %A1=4]]
```

```
(%i196) ydef : at (yundef, solns[1]);
```

```
(ydef) (4.5356 sin(0.72273 t) + 4 cos(0.72273 t)) 0.5^t + 2.0
```

```
(%i197) yd(t) := "ydef";
```

```
(%o197) yd(t) := (4.5356 sin(0.72273 t) + 4 cos(0.72273 t)) 0.5^t + 2.0
```

```
(%i198) tL : makelist (t, t, 0, 8);
```


```
(tL) [0, 1, 2, 3, 4, 5, 6, 7, 8]
```

```
(%i199) YtL : map ('yd, tL);
```

```
(YtL) [6.0, 5.0, 3.25, 2.1875, 1.8281, 1.8242, 1.9111, 1.9773, 2.0052]
```

```
(%i201) tmax : 8$
```

```
wxdraw2d (xlabel = "t", ylabel = "Y[t]", xrange = [0, tmax], yrange = [0, 8],
  title = "Complex Roots: a = 0.5, b = 0.5, IaG0 = 1, Y[0] = 6, Y[1] = 5 ",
  points (tL, YtL), line_width = 1, color = red, key = "Ye = 2", explicit(2, t, 0, tmax))$
```



We clearly have convergence, with some oscillation once the curve crosses  $Y_e = 2$ .

Looking at longer times:

```
(%i202) map ('yd, makelist(t,t,0,20));
```

```
(%o202) [6.0,5.0,3.25,2.1875,1.8281,1.8242,1.9111,1.9773,2.0052,2.0096,
2.0059,2.002,2.0,1.9995,1.9996,1.9998,2.0,2.0,2.0,2.0,2.0]
```

Details in this scenario with declining GDP at the start:

```
(%i205) MAlist (0.5, 0.5, 1, 6, 5, 8)$
```

```
makeLists (%)$
```

```
YtL;
```

```
Ye=2.0
```

```
(%o205) [6,5,3.25,2.1875,1.8281,1.8242,1.9111,1.9773,2.0052]
```

```
(%i207) tmax : 8$
```

```
wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [-1, 8],
```

```
title = " Complex roots: a = 1.5, b = 0.5, IaG0 = 1, Y[0] = 6, Y[1] = 5",
```


```
point_type = filled_circle, key = "Yt",points(tL,YtL),color = red,key = "Ct",
```

```
points (tL, CtL), color = brown,key = "Idt",points(tL,IdtL),
```

```
color = magenta, line_width = 1,key = sconcat ( "Ye = ", Ye), explicit(Ye, t, 0, tmax),
```

```
color = black, key = "", explicit (0, t, 0, tmax))$
```

```
(%t207)
```



Details with a rising GDP at the start:

```
(%i210) MAlist (0.5, 0.5, 1, 0.5, 0.7, 8)$
```


```
makeLists (%)$
```

```
YtL;
```

```
Ye=2.0
```

```
(%o210) [0.5,0.7,1.4,1.875,2.0562,2.0734,2.041,2.0124,1.999]
```

```
(%i212) tmax : 8$
wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [-0.25, 3.5], key_pos = top_left,
title = "Complex roots: a = 1.5, b = 0.5, IaG0 = 1, Y[0] = 0.5, Y[1] = 0.7",
point_type = filled_circle, key = "Yt",points(tL,YtL),color = red,key = "Ct",
points (tL, CtL), color = brown,key = "Idt",points(tL,IdtL),
color = magenta, line_width = 1,key = sconcat ( "Ye = ", Ye),
explicit(Ye, t, 0, tmax), color = black, key = "", explicit (0, t, 0, tmax))$
```



(%t212)

With a rising GDP at the start, we again see convergence to  $Y_e$ .


For a convergent complex root case in which  $a > 1$ , we take  $a = 1.5 = 3/2$ ,  $b = 0.5 < 1/a$ ,  $Y[0] = 6$ ,  $Y[1] = 5$ ,  $IaG0 = 1$ ,  $Y_e = 1/(1-b) = 2$ .

```
(%i215) MAlist (1.5, 0.5, 1, 6, 5, 30)$
makeLists (%)$
YtL;
Ye=2.0
```

```
(%o215) [6,5,2.75,0.6875,-0.20313,0.23047,1.4404,2.6277,3.2043,
3.0346,2.39,1.7116,1.347,1.4,1.7398,2.1247,2.3511,2.3453,2.1683,
1.9514,1.813,1.8027,1.8936,2.015,2.0985,2.1119,2.066,1.9986,
1.9487,1.937,1.9597]
```

```
(%i217) tmax : 30$
```

```
wxdraw2d (xlabel = "t", ylabel = "Y[t]", xrange = [0, tmax], yrange = [-1, 8],
  title = "Complex Roots: a = 1.5, b = 0.5, IaG0 = 1, Y[0] = 6, Y[1] = 5 ",
  points (tL, YtL), line_width = 1, color = red,
  key = sconcat ( "Ye = ", Ye), explicit(Ye, t, 0, tmax), color = black, key = "",
  explicit (0, t, 0, tmax))$
```



```
(%t217)
```

## 10.5.2 Borderline Case: Eternal Oscillation

Consider a borderline case in which  $a = 2$ ,  $b = 0.5$ ,  $IaG0 = 10$ ,  $Y[0] = 10$ ,  $Y[1] = 25$ . We call this "borderline" because  $a > 1$  and  $b = 1/a$ , on the border between convergence and divergence. Since  $a*b = 1$ , we have  $k = 1$ ,  $k^t = 1$  for  $t = 0, 1, 2, \dots$  and we get a purely sinusoidal expression. We are taking  $IaG0 = 10$  just for some variety.

```
(%i220) MAlist (2, 0.5, 10, 10, 25, 40)$
```

```
makeLists (%)$
```


```
YtL;
```

```
Ye=20.0
```

```
(%o220) [ 10, 25, 37.5, 41.25, 34.375, 20.313, 6.0938, -1.1719, 2.1484,
  14.395, 29.443, 39.771, 40.212, 30.548, 15.61, 2.8665, -1.31, 5.1685,
  19.063, 33.426, 41.076, 38.188, 26.206, 11.121, 0.47588, -0.40753,
  8.9128, 23.777, 36.752, 41.352, 35.275, 21.561, 7.0665, -0.96143,
  1.4914, 13.198, 28.306, 39.261, 40.585, 31.617, 16.84 ]
```


```
(%i222) tmax : 40$
wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [- 10, 80],
title = "Complex roots: a = 2, b = 1/2, k = 1, IaG0 = 10, Y[0] = 10, Y[1] = 25",
point_type = filled_circle, key = "Yt", points(tL,YtL),
color = magenta, line_width = 1,key = sconcat ( "Ye = ", Ye),
explicit(Ye, t, 0, tmax), color = black, key = "", explicit (0, t, 0, tmax))$
```

(%t222)



```
(%i224) tmax : 40$
wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [-30, 80],
title = "Complex roots: a = 2, b = 1/2, k = 1, IaG0 = 10, Y[0] = 10, Y[1] = 25",
point_type = filled_circle, key = "Yt",points(tL,YtL),color = red,key = "Ct",
points (tL, CtL), color = brown,key = "Idt",points(tL,IdtL),
color = magenta, line_width = 1,key = sconcat ( "Ye = ", Ye),
explicit(Ye, t, 0, tmax), color = black, key = "", explicit (0, t, 0, tmax))$
```


(%t224)



### 10.5.3 Case 3B: $a > 1$ , $1/a < b < 4*a/(1+a)^2$ , divergence

We draw a plot in the (a,b) plane showing this region of oscillating divergence where  $k = \sqrt{a*b} > 1$ .

```
(%i225) wxdraw2d(xlabel = "a",ylabel = "b",key = "4*a/(1+a)^2",
  explicit (4*a/(1+a)^2,a,0,1),color = red,key = "1/a",
  explicit (1/a,a,0.5, 1),filled_func = true, fill_color = dark_gray,
  key = "complex roots, k > 1", explicit (4*a/(1+a)^2,a,1,5),
  fill_color = light_gray, key = "", explicit(1/a,a,1,5), filled_func = false,
  color = blue, explicit (4*a/(1+a)^2,a,1,5), color = red, explicit(1/a,a,1,5),
  color = black, line_width = 1,explicit(1,a,0,5),parametric(1,yy,yy,0,1))$
```



We need a different function to survey values of  $k = \sqrt{a*b}$  for a divergent set (a,b).

```
(%i226) surveyComplex2(%a) :=
  block ([blist, bstart,bend,delb],
    if %a <= 1 then print("%a must be greater than 1")
    else (
      bstart : 1/%a + 1e-10,
      bend : 4*%a/(1+%a)^2 - 1e-10,
      delb : (bend - bstart)/10,
      /* print("bstart = ",bstart,"bend = ", bend," delb = ",delb), */
      blist : makelist (bstart + n*delb,n, 0,10),
      print (" [b, k]"),
      for %b in blist do print ( [%b, sqrt (%a*%b)] ),
      done)$
```

We try this out for  $a = 1.5$ .



```
(%i228) fpprintprec : 16$
surveyComplex2(1.5);
[b, k]
[0.6666666667666666, 1.000000000075]
[0.69600000008, 1.021763182014306]
[0.7253333333933333, 1.043072384875565]
[0.7546666667066666, 1.063954886289828]
[0.7840000000199999, 1.084435336951909]
[0.8133333333333332, 1.104536101718726]
[0.8426666666466667, 1.124277545791074]
[0.87199999996, 1.143678276413433]
[0.9013333332733333, 1.162755348261189]
[0.9306666665866666, 1.181524438968572]
[0.9599999999, 1.1999999999375]
```

```
(%o228) done
```

Experimenting with `surveyComplex2(a)` shows that with

$$a > 1 \text{ and } 1/a < b < 4*a/(1+a)^2$$

one has a complex roots case in which  $k > 1$ , hence a divergent solution, since the complementary solution has the form

$$k^t [A1 \cos(th*t) + A2 \sin(th*t)]$$

with real adjustable constants  $A1$  and  $A2$ .


As a complex root example showing divergence we choose  $a = 2$ ,  $b = 0.75$  (see plot above),  $k = 1.225$ ,  $laG0 = 1$ ,  $Y[0] = 6$ ,  $Y[1] = 5$ .

```
(%i232) fpprintprec : 5$
MAlist (2, 0.75, 1, 6, 5, 30)$
makeLists (%)$
YtL;
Ye=4.0
(%o232) [6, 5, 3.25, 0.8125, -2.0469, -4.8242, -6.7842, -7.0281, -
4.6369, 1.1091, 10.451, 22.851, 36.738, 49.384, 57.007, 55.191, 39.668,
7.4665, -41.702, -104.03, -170.51, -226.61, -253.1, -228.57, -
133.62, 43.202, 298.64, 608.13, 921.34, 1161.8, 1233.1]
```

```
(%i234) tmax : 30$
```

```
wxdraw2d (xlabel = "t", ylabel = "Y[t]", xrange = [0, tmax], yrange = [-400, 1300],  
title = " Complex: a = 2, b = 0.75, k = 1.225 ", key_pos = top_left,  
points (tL, YtL), line_width = 1, color = red, key = sconcat ( "Ye = ", Ye),  
explicit(Ye, t, 0, tmax), color = black, key = "", explicit (0, t, 0, tmax))$
```

```
(%t234)
```




This diverging solution blows right past  $Ye = 4$ , with each step larger than the previous step.

```
(%i236) tmax : 30$
```

```
wxdraw2d(xlabel = "t", xrange = [0, tmax], yrange = [-400, 1300], key_pos = top_left,
title = "Complex roots: a = 2, b = 0.75, k = 1.2, IaG0 = 1, Y[0] = 6, Y[1] = 5",
point_type = filled_circle, key = "Yt", points(tL, YtL), color = red, key = "Ct",
points(tL, CtL), color = brown, key = "Idt", points(tL, IdtL),
color = magenta, line_width = 1, key = sconcat("Ye = ", Ye),
explicit(Ye, t, 0, tmax), color = black, key = "", explicit(0, t, 0, tmax))$
```

```
(%t236)
```



which shows explosive instability.

## 10.6 Summary Plot : Stable and Unstable Regions

We draw a summary plot somewhat similar to Fig. 18.2 in Chiang and Wainwright (except that we use different colors rather than different fill styles to distinguish different regions).

We draw two curves:  $b$  vs  $a$  for which  $a*b = 1$ , and  $b$  vs  $a$  for which  $b = 4*a/(1+a)^2$ . In Maxima's `wxdraw2d` we use `filled_func = true` and `fill_color = light_green`, for example, to paint a region beneath a function with light green color. Regions can then be painted over by changing the choice of color (including white or gray if needed) to end up with different colors for different regions.

The following two calls to solve allow one to design straight lines of the form  $y = a*x + b$  which can then be used to relate labels to curves.

```
(%i237) solve ([1.1=3.6*a + b, 0.75 = 3*a + b], [a, b]), numer;
```

```
(%o237) [[a=0.58333, b=-1]]
```


```
(%i238) solve ([1.65=a1 + b1, 1.3 = 0.8*a1 + b1], [a1, b1]), numer;
```

```
(%o238) [[a1=1.75,b1=-0.1]]
```

```
(%i240) kill(a,b)$
```

```
wxdraw2d (xlabel = "a (acceleration)",ylabel = "b (propensity to consume)",
  xrange = [0,2],
  color = dark_green, explicit (1/a, a, 0.5, 1), key = "Stable",
  fill_color = light_green, filled_func = true, explicit (1,a,0,1),
  fill_color = light_blue, key = "Unstable",explicit (1, a, 1, 5),
  fill_color = light_salmon, key = "Explosively Unstable", explicit (4*a/(1+a)^2, a,1, 5),
  fill_color = dark_gray, key = "Damped Stepped Fluctuation",
  explicit (4*a/(1+a)^2, a, 0, 1), key = "", explicit (1/a, a, 1, 5),
  filled_func = false, color = black, line_width = 1,explicit (1, a, 0, 5),
  parametric (1, yy, yy, 0, 1),
  label ("b = 1/a", 1, 1.7)), label ([ "b = 4*a/(1+a)^2", 4, 1.2]),
explicit (1.75*x -0.1,x, 0.8, 1), explicit (0.583*x -1, x, 3.05, 3.65),
  line_width = 2, color = blue, explicit (1/a, a, 1, 5),
  color = red, explicit (4*a/(1+a)^2, a, 0, 5))$
```

```
(%t240)
```



The stable green region has  $0 < a < 1$  and  $b > 4*a/(1+a)^2 > 0$ , or  $a*b > 4*a^2/(1+a)^2$ .

## 10.7 Economic Instability with the Acceleration Principle

Quoting from Ch. 17, "The Business Cycle", p. 405, Economics, by Paul A. Samuelson, 1st. ed., 1948, McGraw-Hill:

"The Acceleration Principle:

According to this law, society's needed stock of capital, whether inventory or equipment, depends primarily upon the level of income or production. Additions to the stock of capital, or what we customarily call net investment [induced investment], will take place only when income is growing. As a result, a prosperity period may come to an end - not simply because consumption sales have gone down - but simply because sales have leveled off at a high level or have continued to grow, but at a lower rate than previously."

"A simplified arithmetical example will make this clear. Imagine a typical textile-manufacturing firm whose stock of capital equipment is always kept equal to about ten times the value of its yearly sales of cloth. Thus, when its sales have remained at 6 million dollars per year for some time, its balance sheet will show 60 million dollars of capital equipment, consisting of perhaps 20 machines of different ages with one wearing out each year and being replaced. Because replacement balances depreciation, there is no "net" investment or saving being done by the corporation. "Gross" investment is taking place at the rate of 3 million dollars per year, representing the yearly replacement of one machine. (The other 3 million dollars of sales may be assumed to go for wages and dividends.) [Gross investment (G.I.) is the sum of net investment (N.I.) plus replacement.]"

Let us call the first three years (sales = 6 million/year) "Phase 1", the fourth, fifth, and sixth years (sales = 9, 12, 15 million/yr) "Phase 2", the seventh year "Phase 3", and the eighth year "Phase 4".

"... suppose that, in the fourth year, sales [of cloth] rise by 50 per cent - from 6 to 9 million dollars. Then the number of machines must also rise by 50 per cent, or from 20 to 30 machines. In the fourth year, 11 machines must be bought, 10 new ones in addition to the replacement of the worn-out one."

"Sales have gone up by 50 per cent. How much has machine production gone up? By 1000 per cent! [ the fractional increase =  $(11 - 1)/1$ , times 100% ] "

"It is this accelerated effect of a change in consumption [of cloth] on [net] investment levels that gives the acceleration principle its name."

"If sales continue to rise in [each of] the fifth and sixth years by 3 million dollars [per year], then we shall continue to have 11 new machines ordered in every year [of Phase 2]."

"So far the acceleration principle has given us no trouble. On the contrary, it has given us a tremendous increase in investment spending as a result of a moderate increase in consumption sales. But now we are riding a tiger. Consumption has to keep increasing in order for investment to stand still! If consumption should stop growing at so rapid a rate - if it should level off in the seventh year even at the high level of 15 million dollars per year - then net investment will fall away to zero, and gross investment will fall back to 1 machine per year (see the table). In other words, a drop of zero per cent in sales has resulted in a 90 per cent drop in gross investment and a 100 per cent drop in net investment!"

$$\Delta(\text{N.I.}) = (0 - 30)/30 \text{ times } 100\% = - 100\%$$

$$\Delta(\text{G.I.}) = (3 - 33)/33 \text{ times } 100\% = - 91\%$$

"The Lord giveth and the Lord taketh away. The acceleration principle is a two-edged sword. If sales should drop below 15 million dollars per year, gross investment would drop away to nothing; in fact, the firm would want to disinvest (N.I. < 0) by selling off some of its machinery [or not replacing a 20 yr old machine no longer needed] on the used-equipment market. It is now clear that a depression can set in just because consumption has stopped growing so rapidly, even if it has not dropped off absolutely, but only leveled off at a high level."

"Needless to say, the curtailment of production in the machine-producing industries will cause them to shut down, will curtail their income and spending on food and clothing, and lead to still further 'multiplier' changes in spending. This might ultimately cause textile sales to stop growing altogether, or even to decline. This will cause a further accelerated drop in net investment. Thus, we may be in a vicious circle whereby the acceleration principle and the multiplier interact so as to produce a cumulative deflationary (or inflationary) spiral."

"It is easy to see that in the acceleration principle we have a powerful factor making for economic instability."

....

"If business sales go up and down, the acceleration principle intensifies their fluctation. It induces net investment on the upswing but causes about the same amount of net disinvestment on the downswing. In the long run, if the system is growing because of population increase or higher real incomes, then the acceleration principle works primarily as a stimulating factor."

...

"The economic system is more or less without a steering wheel. There is no mechanism or automatic governor which keeps purchasing power right at the full-employment level, neither too low nor too high. On the contrary, even if businessmen and workers are doing their best and acting intelligently and unselfishly, the system may still be in the throes of inflation or deflation, depending upon the chance circumstances of the complex interaction between investment and saving.....This is the challenge to fiscal policy."

Let's make a table using the Maxima matrix function, which will worry about formatting. Each list represents one row of the matrix, and each must have the same number of elements.

```
(%i241) matrix (["time", "Yearly Sales", "Stock of Capital", "N.I.", "G.I."],
["=====", "=====", "=====", "=====", "====="],
["Phase 1", " ", " ", " ", " "],
["First year", 6, 60, 0, 3],
["Second year", 6, 60, 0, 3],
["Third year", 6, 60, 0, 3],
["Phase 2", " ", " ", " ", " "],
["Fourth year", 9, 90, 30, 33],
["Fifth year", 12, 120, 30, 33],
["Sixth year", 15, 150, 30, 33],
["Phase 3", " ", " ", " ", " "],
["Seventh year", 15, 150, 0, 3],
["Phase 4", " ", " ", " ", " "],
["Eighth year", 14.7, 147, -3, -3]);
```

(%o241)

<i>time</i>	<i>Yearly Sales</i>	<i>Stock of Capital</i>	<i>N.I.</i>	<i>G.I.</i>
=====	=====	=====	=====	=====
<i>Phase 1</i>				
<i>First year</i>	6	60	0	3
<i>Second year</i>	6	60	0	3
<i>Third year</i>	6	60	0	3
<i>Phase 2</i>				
<i>Fourth year</i>	9	90	30	33
<i>Fifth year</i>	12	120	30	33
<i>Sixth year</i>	15	150	30	33
<i>Phase 3</i>				
<i>Seventh year</i>	15	150	0	3
<i>Phase 4</i>				
<i>Eighth year</i>	14.7	147	-3	-3